** Clearing Stata memory
capture log close
clear all
set more off, perm
set seed 1234

///////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////
//////////////////////////// Table O.24: Priority Subjects and Gender Performance Gap: Preparatory course /////////////////
///////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////

*** Creating variables
encode subject, gen (sub)
tab subject, gen (d_sub)
label var sub "Subject"

** Subject dummies
rename d_sub1 Biology
rename d_sub2 Chemistry
rename d_sub3 Geography
rename d_sub4 History
rename d_sub5 Language
rename d_sub6 Mathematics
rename d_sub7 Physics
rename d_sub8 Portuguese
* Labels
label var Biology "Biology"
label var Chemistry "Chemistry"
label var Geography "Geography"
label var History "History"
label var Math "Mathematics"
label var Physics "Physics"
label var Portuguese "Portuguese"
label var Language "Foreign Language"

** Interaction: priority X female
gen fem_priority=female*priority
label var fem_priority "Female $\times$ Priority"

** Interaction: priority X subject
foreach v of varlist Biology-Portuguese {
gen fem_`v'=`v'*female
label var fem_`v' "Female $\times$ `v'"
gen prio_`v'=priority*`v'
label var prio_`v' "Priority $\times$ `v'"
gen fem_prio_`v'=fem_priority*`v'
label var fem_prio_`v' "Female $\times$ Priority $\times$ `v'"
}

global subject "Chemistry Geography History Mathematics Physics"
global subject_fem "fem_Chemistry fem_Geography fem_History fem_Mathematics fem_Physics"

** P1 scores: P1 normalized subject-specific scores
forvalues i=2(1)4 {
gen norm_p1score`i'=norm_p1score^`i'
sum norm_p1score`i'
}

*********************************************************************************
****************   Relative performances ****************************************
*********************************************************************************

** ENEM

foreach v in  norm_enem_w {
bys year female: egen `v'_ave_g=mean(`v')
gen `v'_g=`v'-`v'_ave_g
bys year female: sum `v'_g
}
drop norm_enem_w_ave_g

* Interaction: subject X ENEM
foreach v of varlist Biology-Portuguese {
gen enem_`v'=`v'*norm_enem_w_g
label var enem_`v' "ENEM $\times$ `v'"
gen fem_enem_`v'=female*norm_enem_w_g*`v'
label var fem_enem_`v' "Female $\times$ ENEM $\times$ `v'"
forvalues i=2(1)4 {
gen enem_`v'_`i'=enem_`v'^`i'
gen fem_enem_`v'_`i'=fem_enem_`v'^`i'
}
sum enem_`v'* fem_enem_`v'*
}

global g_pol_enem_sub "enem_Chemistry* enem_Geography* enem_History* enem_Mathematics* enem_Physics*"
d $g_pol_enem_sub

* Priority x relative performance in ENEM:
foreach v in norm_enem_w {
gen `v'_priority_g=`v'_g*priority
forvalues i=2(1)4 {
gen `v'_priority_g`i'=`v'_g^`i'*priority
sum `v'_priority_g`i'
}
}

global g_norm_enem_w_prio norm_enem_w_priority_g*
d $g_norm_enem_w_prio

** Phase 1 scores

foreach v in norm_p1score {

tab year, sum(`v')
bys year subject female: egen gs_`v'_ave=mean(`v')
gen gs_`v'=`v'-gs_`v'_ave
bys year female subject: sum gs_`v'
drop gs_`v'_ave

forvalues i=2(1)4 {
gen gs_`v'`i'=gs_`v'^`i'
sum gs_`v'`i'
}

global gs_pol_`v' gs_`v' gs_`v'2 gs_`v'3 gs_`v'4
d $gs_pol_`v'

* Priority x Phase 1 scores:
gen gs_`v'_prio=gs_`v'*priority
forvalues i=2(1)4 {
gen gs_`v'_prio`i'=gs_`v'`i'*priority
sum gs_`v'_prio*
}
}

global gs_pol_norm_p1score_prio gs_norm_p1score_prio*
d $gs_pol_norm_p1score_prio

*********************************************************************************
**************** Main sample ****************************************************
*********************************************************************************

* 1) Only years before the affirmative action took place
drop if aa_year==1
drop if year==2000
tab year

* 2) Drop Portuguese and Foreign Language (in Phase 1 there is no Portuguese or Foreign Language exams - For Portuguese Phase 1 has an essay)
 tab subject, sum(norm_p1score)
 drop if subject=="lang" | subject=="port" 
 tab subject, sum(norm_p1score)
 drop Language Portuguese prio_Language prio_Portuguese fem_prio_Language fem_prio_Portuguese 
 
tab prep_course, mi
tab year, sum(prep_course)
drop if prep_course==.

****************************************************************************************
****************  Regressions  *********************************************************
****************************************************************************************

estimates clear

reg norm_score female priority fem_priority norm_enem_w_g if prep_course==1, cluster(inscri2) 
estimates store prepcourse1
reg norm_score female priority fem_priority norm_enem_w_g $subject $subject_fem if prep_course==1, cluster(inscri2) 
estimates store prepcourse2
reghdfe norm_score priority fem_priority $subject $subject_fem if prep_course==1, cluster(inscri2) absorb(inscri2)  
estimates store prepcourse3
reghdfe norm_score priority fem_priority $subject $subject_fem $g_pol_enem_sub if prep_course==1, cluster(inscri2) absorb(inscri2)  
estimates store prepcourse4
reghdfe norm_score priority fem_priority $subject $subject_fem $g_pol_enem_sub $g_norm_enem_w_prio if prep_course==1, cluster(inscri2) absorb(inscri2)  
estimates store prepcourse5
reghdfe norm_score priority fem_priority $subject $subject_fem $g_pol_enem_sub $g_norm_enem_w_prio $gs_pol_norm_p1score if prep_course==1, cluster(inscri2) absorb(inscri2)  
estimates store prepcourse6
reghdfe norm_score priority fem_priority $subject $subject_fem $g_pol_enem_sub $g_norm_enem_w_prio $gs_pol_norm_p1score $gs_pol_norm_p1score_prio if prep_course==1, cluster(inscri2) absorb(inscri2)  
estimates store prepcourse7

reg norm_score female priority fem_priority norm_enem_w_g if prep_course==0, cluster(inscri2) 
estimates store noprepcourse1
reg norm_score female priority fem_priority norm_enem_w_g $subject $subject_fem if prep_course==0, cluster(inscri2) 
estimates store noprepcourse2
reghdfe norm_score priority fem_priority $subject $subject_fem if prep_course==0, cluster(inscri2) absorb(inscri2)  
estimates store noprepcourse3
reghdfe norm_score priority fem_priority $subject $subject_fem $g_pol_enem_sub if prep_course==0, cluster(inscri2) absorb(inscri2)  
estimates store noprepcourse4
reghdfe norm_score priority fem_priority $subject $subject_fem $g_pol_enem_sub $g_norm_enem_w_prio if prep_course==0, cluster(inscri2) absorb(inscri2)  
estimates store noprepcourse5
reghdfe norm_score priority fem_priority $subject $subject_fem $g_pol_enem_sub $g_norm_enem_w_prio $gs_pol_norm_p1score if prep_course==0, cluster(inscri2) absorb(inscri2)  
estimates store noprepcourse6
reghdfe norm_score priority fem_priority $subject $subject_fem $g_pol_enem_sub $g_norm_enem_w_prio $gs_pol_norm_p1score $gs_pol_norm_p1score_prio if prep_course==0, cluster(inscri2) absorb(inscri2)  
estimates store noprepcourse7

estadd local sub_fe "No":  prepcourse1 noprepcourse1
estadd local sub_fe "Yes": prepcourse2 prepcourse3 prepcourse4 prepcourse5 prepcourse6 prepcourse7 noprepcourse2 noprepcourse3 noprepcourse4 noprepcourse5 noprepcourse6 noprepcourse7

estadd local subgender_fe "No":  prepcourse1 noprepcourse1
estadd local subgender_fe "Yes": prepcourse2 prepcourse3 prepcourse4 prepcourse5 prepcourse6 prepcourse7 noprepcourse2 noprepcourse3 noprepcourse4 noprepcourse5 noprepcourse6 noprepcourse7

estadd local ind_fe "No": prepcourse1 prepcourse2 noprepcourse1 noprepcourse2 
estadd local ind_fe "Yes": prepcourse3 prepcourse4 prepcourse5 prepcourse6 prepcourse7 noprepcourse3 noprepcourse4 noprepcourse5 noprepcourse6 noprepcourse7

estadd local enemsub "No":  prepcourse1 prepcourse2 prepcourse3 noprepcourse1 noprepcourse2 noprepcourse3 
estadd local enemsub "Yes": prepcourse4 prepcourse5 prepcourse6 prepcourse7 noprepcourse4 noprepcourse5 noprepcourse6 noprepcourse7

estadd local enemprio_pol4 "No":  prepcourse1 prepcourse2 prepcourse3 prepcourse4 noprepcourse1 noprepcourse2 noprepcourse3 noprepcourse4 
estadd local enemprio_pol4 "Yes": prepcourse5 prepcourse6 prepcourse7 noprepcourse5 noprepcourse6 noprepcourse7

estadd local p1score_pol4 "No": prepcourse1 prepcourse2 prepcourse3 prepcourse4 prepcourse5 noprepcourse1 noprepcourse2 noprepcourse3 noprepcourse4 noprepcourse5
estadd local p1score_pol4 "Yes": prepcourse6 prepcourse7 noprepcourse6 noprepcourse7

estadd local p1scoreprio_pol4 "No": prepcourse1 prepcourse2 prepcourse3 prepcourse4 prepcourse5 prepcourse6 noprepcourse1 noprepcourse2 noprepcourse3 noprepcourse4 noprepcourse5 noprepcourse6
estadd local p1scoreprio_pol4 "Yes": prepcourse7 noprepcourse7 

* Comparing coefficients
reghdfe norm_score i.prep_course##c.(priority fem_priority $subject $subject_fem $g_pol_enem_sub $g_norm_enem_w_prio $gs_pol_norm_p1score $gs_pol_norm_p1score_prio) , cluster(inscri2) absorb(inscri2)
local pval = (2 * ttail(e(df_r), abs(_b[1.prep_course#c.fem_priority] / _se[1.prep_course#c.fem_priority]) ) )
display `pval'
estimates restore prepcourse1
estadd scalar p_value_coefs = `pval'
estimates store prepcourse1

* Tex
esttab prepcourse* noprepcourse* using "Output/p_prep_course.tex", se star(* 0.10 ** 0.05 *** 0.01) nogap ///
stats(p_value_coefs line r2_a N N_clust sep sub_fe subgender_fe ind_fe enemsub  enemprio_pol4 p1score_pol4 p1scoreprio_pol4 , fmt(%9.3fc %1s %9.3fc %9.0fc %9.0fc %1s %3s %3s %3s %3s %3s %3s %3s %3s) labels("P-value (Female $\times$ Priority)" " " "$\bar{R}^2$" "Number of observations"  "Number of applicants" " " "Subject FE" "Subject-gender FE" "Individual FE" "ENEM $\times$ Subject FE" "ENEM $\times$ Priority" "Phase 1 scores"  "Phase 1 scores $\times$ Priority"  )) b(%7.3f) se(%7.3f)  booktabs replace f label nomtitle collabels(none) keep(female priority fem_priority norm_enem_w_g) mgroups("Preparatory course" "No preparatory course", pattern (1 0 0 0 0 0 0 1 0 0 0 0 0 0) prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span})) ///
 refcat(female " \\ \multicolumn{15}{l}{\textit{Dependent variable: Phase 2 normalized subject-specific scores}} \\", nolabel)

**